
At the conclusion of this unit, you will be able to:

Logical Database

: 

l

l

-

Read data with a logical database.

· In general, the system reads data that will appear in a list from the database.

· You can use either OPEN SQL or NATIVE SQL statements to read data from the database. Native SQL should be avoided in order to maintain portability from one RDBMS to another. In addition, Native SQL would need to be recoded to take advantage of upgrades in the RDBMS. OPEN SQL has no such limitations or requirements.

· The use of a logical database program provides you with an excellent alternative to programming the data retrieval yourself.  Logical databases retrieve data records in primary key sequence and make them available to ABAP programs.They also offer a security advantage since the authority checks are done in the logical database program rather than in the report program.

Advantages of a Logical Database


· The same logical database can be the data source for several QuickViews, queries, and reports. In the QuickView, the logical database program (LDB) can be specified directly as a data source. A query works with the logical database when the InfoSet that generated the query is defined with a logical database. In the case of executable programs, the LDB is entered in the attributes or called using function module LDB_PROCESS. See appendix for information on how to use the function module.

· Logical databases offer several advantages over other methods of data retreival:

· The system generates a selection screen. The use of selection screen versions or variants provides the required flexibility. 

· The programmer does not need to know the exact structure or relationship of the tables involved; the data is made available to the application program in primary key sequence. Data is formatted for output in the GET events. 

· Performance improvements within logical databases directly affect all programs linked to the logical database without having to change the programs themselves.  

· Maintenance and enhancements can be performed at a central location.

· Authorization checks are done in the logical database program rather than in the application program. 

Logical Database:

Overview


· A logical database is an ABAP program that reads data from the database and makes the data available to other report programs. 

· A hierarchical structure determines the order in which the data is supplied to the programs. A logical database also provides a selection screen that checks user entries and conducts error dialogs. You can add your own parameters and select options.     

· The SAP system provides approximately 190 logical databases in Release 4.6. The names of logical databases have been extended from three characters to 20 characters in Release 4.0 (namespace prefix max. 10 characters).

· If the logical database is three characters long, then the first two characters name the logical database and the third character identifies the application area. A list of the values for the third character can be displayed by looking at a program's attributes and then pressing F4 on the Application field. For example, DDF is named for Debtor Data and the letter F represents financial data.

Logical Database:

F1S Nodes

CARRID

CONNID

FLDATE

BOOKID

Flight schedule

MANDT

CARRID

CONNID

COUNTRYFR        

CITYFROM         

AIRPFROM         

COUNTRYTO        

CITYTO           

AIRPTO           

FLTIME           

DEPTIME          

ARRTIME          

DISTANCE         

DISTID           

FLTYPE           

PERIOD           

SBOOK

SPFLI

SFLIGHT

Flights

MANDT

CARRID

CONNID

FLDATE

PRICE            

CURRENCY

PLANETYPE        

SEATSMAX         

SEATSOCC         

PAYMENTSUM       

CUSTOMID

CUSTTYPE          

SMOKER            

LUGGWEIGHT        

WUNIT             

INVOICE           

CLASS             

FORCURAM          

FORCURKEY         

LOCCURAM          

LOCCURKEY         

ORDER_DATE        

COUNTER           

AGENCYNUM         

· The demo programs and exercises for ABAP courses and ABAP documentation refer to the SAP system's BC_TRAVEL flight data model, which is found in development class BC_DATAMODEL.

· The following tables form the nodes of logical database F1S:

· SPFLI: Flight connections

· SFLIGHT: Flights

· SBOOK: Bookings

[image: image7.wmf]ã

SAP

AG 2002

REPORT sapbc405_ldbd_simple_example.

NODES:

spfli, sflight.

* Processing of SPFLI records     

GET spfli FIELDS ...

.      

* Processing of SFLIGHT records   

GET sflight FIELDS ...

.        

Sample Program for a Logical Database

ABAP:

Program Attributes

Logical database

Selection screen version

F1S


· Only executable or report programs can use logical databases. You must enter the logical database name in the program's attributes.

· Use the keyword NODES <node> to specify the nodes of the logical database that you want to use in the program.  NODES allocates the appropriate storage space for the node (a work area or a table area depending on the node type). Note that the keyword TABLES can still be used in release 4.6.

· When the logical database reads the data records, it issues the keyword PUT <node> which triggers the GET event in the report program. It is important to note that the GET <node> does not retrieve the data. It is simply an event like any other. As with other ABAP event keywords, the sequence in which these events are processed is determined by the structure of the logical database.

· It is important that you code the get event using the FIELDS extension.  GET <node> is the equivalent of SELECT * which is unacceptable in terms of performance. 

Type that was defined in a type group.

Dynamic type

LDB Sub

-

Objects:

SBOOK

Table

Timetable

Name of node

Node type

Short text


· Logical databases are made up of several subobjects. The structure determines the hierarchy, and thus the read sequence of the data records.

· Node names can contain up to 14 characters. There are four different node types:

· Table (type T): The node name is the name of a transparent table (this type corresponds to the concept prior to Release 4.0A). The table name must be identical to the node name. Deep types (complex) are not allowed.                                                             

· DDIC type (type S): Any node name is possible. The node is assigned a structure or a table type from the Dictionary. The node name can differ from the type name. Deep structures are possible. 

· Type groups (type C): The node type is defined in a type group. The name of the type group must be maintained in the Type group field. You should generally prefer DDIC types since the other applications that use the logical database (such as SAP Query) can access them (short texts, and so on).            

· Dynamic nodes (type A): These nodes do not have a fixed type; they are not classified until the program runtime. The types that are allowed are determined when the structure is created.                                                                        

Events in Logical Databases

SBOOK

SFLIGHT


· The START-OF-SELECTION event triggers before control is given to the read routine of the logical database. The END-OF-SELECTION event triggers after all GET events have been processed (all data records have been read and processed).

· The GET <node> event triggers whenever the logical database supplies data for this node. This means that GET events are processed several times, and that data has already been read from the database for these events. The sequence in which the GET events are processed is determined by the structure of the logical database.

· The GET <node> LATE event triggers when all subordinate nodes of node <node> have been processed, before the data is read for the next <node> (whenever a hierarchy level has been completed).

· At the start of the event, the system automatically adds a line feed and sets the font and intensity to the system defaults. If colors were set on in other events, they are reset as well.

· The CHECK keyword ends the current processing block if the condition is false. 

· The STOP keyword ends program processing. However, in contrast to the EXIT statement, the processing block END-OF-SELECTION processes first (if it exists).

· If there is a STOP statement within the END-OF-SELECTION processing block, program processing ends and the list buffer is displayed.

· The EXIT statement exits the program but does not do the END-OF-SELECTION event. Any list buffer created before the EXIT keyword is encountered will be displayed.

· You can also use the REJECT statement. The data record does not process further. Processing continues on the same hierarchy level when the next data record is read. REJECT can also be used within a subroutine.                         

· Logical database programs have an include named db<name>sel in which the logical database selection screen(s) are defined. The addition FOR NODE assigns selections to individual logical nodes. The appearance of a selection screen directly depends on the NODES statement contained within your program. 

· A field selection can be defined in the logical database program for the individual nodes. This is done with the keyword FIELD SELECTION FOR NODE in the SELECTION-SCREEN statement. This code in the logical database program allows the application program to use the GET <node> FIELD <field list> to restrict the amount of data retrieved. 

· Some logical databases also designate dynamic selections for individual nodes. This is done by dynamic selection using the addition DYNAMIC SELECTIONS FOR NODE.  Dynamic selection button then appears on your report's selection screen. You can determine which selection fields can be set by choosing a particular selection view yourself (type: CUS) or by using the selection view delivered by the SAP system (type: SAP). 

· Some large, logical databases have several selection screen versions. Each selection screen version contains a subset of the selection criteria (language element: EXCLUDE). You can specify which selection screen to display by entering the screen number in the program attributes.

· When you enter a logical database in the attributes of your type 1 program, the system processes the selection screen of the logical database. The concrete characteristics of the selection screen depend upon the node specified in the NODES statement. If you specify a node of type T (table), you can also declare the table work area with the TABLES statement. 

· If you address only subordinate nodes (in the hierarchy) of the logical database in the program (for example, SFLIGHT), the selection screen criteria for the superior node in the hierarchy (SPFLI) also appears. You can thus restrict the dataset to be read so that it meets your specific requirements. 

· Note: A logical database always reads in accordance with its structure. This means that if you only need data from a node deep in the hierarchy, you should either find a different logical database where the desired node is higher in the structure or program the data retrieval yourself.  

· If the logical database supports dynamic selections, Dynamic selections appears on the selection screen. When the user selects Dynamic selections, a second selection screen is displayed. This screen allows the user to select additional database fields. The system transfers the selections directly to the logical database program and to the database (dynamic selections).

· The selection view determines which fields are displayed on the selection screen. Create your own view with type CUS, and have it override the view with type SAP.

· The logical database programs are named SAPDB<ldbname> for logical database <ldbname> .The programs are a collection of subroutines, each of which is performed for specific events. For example, subroutine <init> is processed once at the start of the database program. This program can define default values for the selection screen of the LDB. 

· Other subroutines also exist that are processed during events PBO (Process Before Output) and PAI (Process After Input) of the selection screen. Checks, such as authorization checks (AUTHORITY-CHECK), are usually performed during event PAI.

· The database accesses (SELECT statements) are programmed in the put_<node> subroutines. These subroutines may be processed several times, depending on which selection criteria the user specifies. The sequence in which these subroutines are processed is determined by the structure of the logical database. 

SFLIGHT

SBOOK

SELECT * FROM SPFLI ...

PUT SPFLI.

ENDFORM.

FORM PUT_SFLIGHT.

SELECT * FROM SFLIGHT .

PUT SFLIGHT.

ENDSELECT.

ENDFORM.

FORM PUT_SBOOK.

SELECT * FROM SBOOK ...

PUT SBOOK.

ENDSELECT.

ENDFORM.

Structure of F1S

Structure of F1S

* Processing of SPFLI records     

* Processing of SFLIGHT records   

* Processing of SBOOK records   

PUT SBOOK.

PUT SBOOK.

PUT SPFLI.

PUT SPFLI.

PUT SFLIGHT.

PUT SFLIGHT.

GET SBOOK FIELDS ...

GET SBOOK FIELDS ...

GET SPFLI FIELDS ...

GET SPFLI FIELDS ...

GET SFLIGHT FIELDS ...

GET SFLIGHT FIELDS ...

PROGRAM SAPDBF1S

DEFINING DATABASE F1S.

...

Data processing

Data processing

Data processing

Data processing


· When a report program that uses a logical database is started, control is initially passed to the database program of the logical database. Each event has a corresponding subroutine in the database program (for example, subroutine init for event INITIALIZATION). During the interaction between the LDB and the associated report, the subroutine is always processed first, followed by the event (if there is one in the report).

· Logical database programs read data from a database according to the structure declared for the logical database. They begin with the root node and then process the individual branches consecutively from top to bottom. 

· The logical database reads the data in the PUT_<NODE> subroutines. During event PUT, control passes from the database program to the GET event of the associated report. The data is made available in the corresponding work areas in the report. The processing block defined for the GET event is performed. Control then returns to the logical database. PUT activates the next form subroutine found in the structure. This flow continues until the report collects all the available data.

· The depth of data read in the structure depends upon a program's GET events. A logical database reads to the lowest GET event contained within the structure attributes. Only those GET events for which processing is supposed to take place are written into the report program. Logical databases read all data records found on the direct access path. 

GET

sflight

price currency.

CHECK

.

...                                                             

GET sbook FIELDS ...

LDB selections

Checking Internal Program Selections

Dep

. time       08:00:00           23:00:00

Price                  500.00            1000.00

SELECT

SELECT

-

-

OPTIONS:

OPTIONS:

CHECK

CHECK

sprice

sprice

.

.

Negative 

CHECK

Positive 

CHECK

SBOOK

SBOOK

SFLIGHT

SFLIGHT

Are dynamic selections

allowed in this table?

SPFLI                              

YES

SFLIGHT

NO


· If you specify a logical database and declare additional selections in the program attributes that refer to the fields of a node not designated for dynamic selection, you must use the CHECK <seltab> statement to see if the current data record fulfills the selection criteria.

· If the data record does not fulfill these selection criteria, current event block processing ends.

Logical Database: Unit Summary

logical database.

Read data with a logical database. 


Logical Database Exercises

	Unit: Logical Databases

Topic: GET Events

	When you have completed these exercises, you will be able to:

· Create a list with data read from a logical database.




1-1
Create program Z##LDB1_... with TOP include (Z##LDB1_...TOP) and enter logical database F1S in the program attributes. Make sure you specify Executable program as the program type. To do this, you must change the program type. Sample solution for exercise: SAPBC405_LDBS_1. 

1-1-1
The logical database should supply the program with data for nodes SPFLI, SFLIGHT, and SBOOK.

1-1-2
Create a list that displays the following data:

Table SPFLI: CARRID, CONNID, CITYFROM, AIRPFROM, CITYTO, AIRPTO.

Table SFLIGHT: FLDATE, PRICE, CURRENCY, PLANETYPE, SEATSMAX, SEATSOCC, FREE_SEATS.

Table SBOOK: BOOKID, CUSTOMID, SMOKER, LUGGWEIGHT, WUNIT.

Field FREE_SEATS is not a table field – it has to be calculated in the program. The price and luggage weight should be output with the appropriate units.

1-1-3
Create a three line list in which each line outputs the information for one node (see above).

1-1-4
The list should have 83 columns. Maintain the column headers (standard list header).

1-1-5
Format the list (optional)

Output the first line in color COL_HEADING not intensified, the second line in COL_NORMAL intensified, and the third line in COL_NORMAL not intensified. Buffer the list.


Note:
To output the fields, use the pattern functions available in the ABAP Editor.

Exercises

	Unit: Logical Databases

Topic: GET LATE Events and Checks from Internal Program Selections

	When you have completed these exercises, you will be able to:

· Create a list with data read from a logical database.

· Check internal selections for their validity




2-1
Extend program Z##LDB1_... or copy the sample program SAPBC405_LDBS_1 to program Z##LDB2_... . Sample solution for exercise: SAPBC405_LDBS_2.

2-1-1
Add a SELECT-OPTIONS statement for the posting date (SBOOK-ORDER_DATE) to the selection screen. Frame the selection and maintain the selection text.

2-1-2
Make sure that only bookings that meet the specified selection criteria are output in the list. Include the booking date in the list output. 

2-1-3
Maintain the column header (standard list header). In the list, output a solid line when all the bookings for a date have been output and when a flight has been completely output. Display each flight on a new page.

2-2
Optional
Sample solution for exercise: SAPBC405_LDBS_2_OPT.

2-3
Enhance your program.

2-3-1
Add a radio-button group with two radio buttons to the selection screen. Draw a frame around this group and maintain the selection texts. 

The group has to model the following functionality: The user can select between charter flights only and regular flights only in the list. Whether a flight is a charter or not is determined in field SPFLI-FLTYPE: Charter flight: SPFLI –FLTYPE = 'X'.

2-3-2
Make sure that only dates and bookings that meet the specified selection criteria are output in the list. 

Logical Database Solutions

	Unit: Logical Databases

Topic: GET Events


*&---------------------------------------------------------------------*

*& Report  SAPBC405_LDBS_1                                           *

*&                                                                                   *

*&---------------------------------------------------------------------*

*&                                                                     *

*&                                                                     *

*&---------------------------------------------------------------------*

INCLUDE bc405_ldbs_1top.

*&---------------------------------------------------------------------*

*&   Event GET SPFLI

*&---------------------------------------------------------------------*

GET spfli.

* Data output SPFLI

  FORMAT COLOR COL_HEADING INTENSIFIED OFF.

  WRITE:  / sy-vline, spfli-carrid,

            spfli-connid,

            spfli-cityfrom,

            spfli-airpfrom,

            spfli-cityto,

            spfli-airpto,

            AT line_size sy-vline.

*&---------------------------------------------------------------------*

*&   Event GET SFLIGHT

*&---------------------------------------------------------------------*

GET sflight.

* Calculate free seats

  free_seats = sflight-seatsmax - sflight-seatsocc.

* Data output SFLIGHT

  FORMAT COLOR COL_NORMAL INTENSIFIED ON.

  WRITE: / sy-vline, sflight-fldate,

           sflight-price CURRENCY sflight-currency,

           sflight-currency,

           sflight-planetype,

           sflight-seatsmax,

           sflight-seatsocc,

           free_seats,

           AT line_size sy-vline.

*&---------------------------------------------------------------------*

*&   Event GET SBOOK

*&---------------------------------------------------------------------*

GET sbook.

  FORMAT COLOR COL_NORMAL INTENSIFIED OFF.

  WRITE: / sy-vline, sbook-bookid,

           sbook-customid,

           sbook-smoker,

           sbook-luggweight UNIT sbook-wunit,

           sbook-wunit, AT line_size sy-vline.
*&---------------------------------------------------------------------*

*& Include BC405_LDBS_1TOP                                           *

*&                                                                                   *

*&---------------------------------------------------------------------*

REPORT   sapbc405_ldbs_1 LINE-SIZE 83.

* Used nodes of the structure of the logical database F1S

NODES: spfli, sflight, sbook.

* Variables

DATA: free_seats LIKE sflight-seatsocc.

* Constants

CONSTANTS: line_size LIKE sy-linsz VALUE 83.

Solutions

	[image: image23.png]




	Unit: Logical Databases

Topic: GET LATE Events and Checks from Internal Program Selections


*&---------------------------------------------------------------------*

*& Report  SAPBC405_LDBS_2                                           *

*&                                                                                   *

*&---------------------------------------------------------------------*

*&                                                                     *

*&                                                                     *

*&---------------------------------------------------------------------*

INCLUDE bc405_ldbs_2top.

*&---------------------------------------------------------------------*

*&   Event GET SPFLI

*&---------------------------------------------------------------------*

GET spfli.

* Data output SPFLI

  FORMAT COLOR COL_HEADING INTENSIFIED OFF.

  WRITE:  / sy-vline, spfli-carrid,

            spfli-connid,

            spfli-cityfrom,

            spfli-airpfrom,

            spfli-cityto,

            spfli-airpto,

            AT line_size sy-vline.

*&---------------------------------------------------------------------*

*&   Event GET SFLIGHT

*&---------------------------------------------------------------------*

GET sflight.

* Calculate free seats

  free_seats = sflight-seatsmax - sflight-seatsocc.

* Data output SFLIGHT

  FORMAT COLOR COL_NORMAL INTENSIFIED ON.

  WRITE: / sy-vline, sflight-fldate,

           sflight-price CURRENCY sflight-currency,

           sflight-currency,

           sflight-planetype,

           sflight-seatsmax,

           sflight-seatsocc,

           free_seats,

           AT line_size sy-vline.

*&---------------------------------------------------------------------*

*&   Event GET SBOOK

*&---------------------------------------------------------------------*

GET sbook.

* Check select-option

  CHECK so_odat.

  FORMAT COLOR COL_NORMAL INTENSIFIED OFF.

  WRITE: / sy-vline, sbook-bookid,

           sbook-customid,

           sbook-smoker,

           sbook-luggweight UNIT sbook-wunit,

           sbook-wunit,

           sbook-order_date,

           AT line_size sy-vline.

*&---------------------------------------------------------------------*

*&   Event GET SPFLI LATE

*&---------------------------------------------------------------------*

GET spfli LATE.

  ULINE.

  NEW-PAGE.

*&---------------------------------------------------------------------*

*&   Event GET SFLIGHT LATE

*&---------------------------------------------------------------------*

GET sflight LATE.

  ULINE.
*&---------------------------------------------------------------------*

*& Include BC405_LDBS_2TOP                                           *

*&                                                                                   *

*&---------------------------------------------------------------------*

REPORT   sapbc405_ldbs_2 LINE-SIZE 83.

* Used nodes of the structure of the logical database F1S

NODES: spfli, sflight, sbook.

* Additional selections

SELECTION-SCREEN BEGIN OF BLOCK order WITH FRAME.

SELECT-OPTIONS: so_odat FOR sbook-order_date.
SELECTION-SCREEN END OF BLOCK order.

* Variables

DATA: free_seats LIKE sflight-seatsocc.

* Constants

CONSTANTS: line_size LIKE sy-linsz VALUE 83.

Optional Exercises

*&---------------------------------------------------------------------*

*& Report  SAPBC405_LDBS_2_OPT                          *

*&                                                                                   *

*&---------------------------------------------------------------------*

*&                                                                     *

*&                                                                     *

*&---------------------------------------------------------------------*

INCLUDE BC405_LDBS_2_OPTTOP.

*&---------------------------------------------------------------------*

*&   Event GET SPFLI

*&---------------------------------------------------------------------*

GET spfli.

*++++++++++++++++++++++++++++++++> optional

* Check radio button group using a help variable

* Flight type charter or scheduled)

  CLEAR check_negative.

  IF pa_fty1 = 'X'.

    IF NOT spfli-fltype = pa_fty1.

      check_negative = 'X'.

    ENDIF.

  ELSEIF pa_fty2 = 'X'.

    IF NOT spfli-fltype = space.

      check_negative = 'X'.

    ENDIF.

  ENDIF.

  CHECK check_negative = space.

*<---------------------------------------------------------- optional

* Data output SPFLI

  FORMAT COLOR COL_HEADING INTENSIFIED OFF.

  WRITE:  / sy-vline, spfli-carrid,

            spfli-connid,

            spfli-cityfrom,

            spfli-airpfrom,

            spfli-cityto,

            spfli-airpto,

            AT line_size sy-vline.

*&---------------------------------------------------------------------*

*&   Event GET SFLIGHT

*&---------------------------------------------------------------------*

GET sflight.

* Calculate free seats

  free_seats = sflight-seatsmax - sflight-seatsocc.

* Data output SFLIGHT

  FORMAT COLOR COL_NORMAL INTENSIFIED ON.

  WRITE: / sy-vline, sflight-fldate,

           sflight-price CURRENCY sflight-currency,

           sflight-currency,

           sflight-planetype,

           sflight-seatsmax,

           sflight-seatsocc,

           free_seats,

           AT line_size sy-vline.

*&---------------------------------------------------------------------*

*&   Event GET SBOOK

*&---------------------------------------------------------------------*

GET sbook.

* Check select-option

  CHECK so_odat.

  FORMAT COLOR COL_NORMAL INTENSIFIED OFF.

  WRITE: / sy-vline, sbook-bookid,

           sbook-customid,

           sbook-smoker,

           sbook-luggweight UNIT sbook-wunit,

           sbook-wunit,

           sbook-order_date,

           AT line_size sy-vline.

*&---------------------------------------------------------------------*

*&   Event GET SPFLI LATE

*&---------------------------------------------------------------------*

GET spfli LATE.

  ULINE.

  NEW-PAGE.

*&---------------------------------------------------------------------*

*&   Event GET SFLIGHT LATE

*&---------------------------------------------------------------------*

GET sflight LATE.

  ULINE.

*&---------------------------------------------------------------------*

*& Include BC405_LDBS_2OPTTOP                                   *

*&                                                                     *

*&---------------------------------------------------------------------*

REPORT   sapbc405_ldbs_2_opt LINE-SIZE 83.

* Used nodes of the structure of the logical database F1S

NODES: spfli, sflight, sbook.

* Additional selections

SELECTION-SCREEN BEGIN OF BLOCK order WITH FRAME.

SELECT-OPTIONS: so_odat FOR sbook-order_date.

SELECTION-SCREEN END OF BLOCK order.

* Additional selections (optional part)

SELECTION-SCREEN BEGIN OF BLOCK type WITH FRAME.

PARAMETERS: pa_fty1 RADIOBUTTON GROUP ftyp,
                            pa_fty2 RADIOBUTTON GROUP ftyp.

SELECTION-SCREEN: END OF BLOCK type.
* Variables

DATA: free_seats LIKE sflight-seatsocc.

DATA  check_negative.

* Constants

CONSTANTS: line_size LIKE sy-linsz VALUE 83.

